

实验二 软件系统模拟实验

一、实验要求

1、实验之前认真预习，明确实验的目的和具体实验内容，写出实验用到的数据和控制信号的取值，做好实验之前的必要准备。

2、想好实验的操作步骤，明确通过实验到底可以学习哪些知识，想一想怎么样有意识地提高教学实验的真正效果。

3、在教学实验过程中，要爱护教学实验设备和用到的辅助仪表，记录实验步骤中的数据和运算结果，仔细分析遇到的现象与问题，找出解决问题的办法，有意识地提高自己创新思维能力。

4、实验之后认真写出实验报告，重点在于预习时准备的内容，实验数据，运算结果的分析讨论，实验过程、遇到的现象和解决问题的办法，自己的收获体会，对改进教学实验安排的建议等。善于总结和发现问题，写好实验报告是培养实际工作能力非常重要的一个环节，应给以足够的重视。

二、实验目的

1、熟悉教学计算机监控程序的功能。

2、理解计算机软件系统的层次结构，特别是几个层次的语言(机器语言、汇编语言、高级语言)的功能和用法上的同异之处，通过用BASIC语言写几个小的程序，体会高级语言和汇编语言在处理能力和使用的方便程度等方面的区别。

三、实验注意事项

1、教学计算机的软件模拟系统，包括指令层次与系统层次的模拟、微体系结构层次的组合逻辑控制器模拟、微程序控制器模拟 3 部分内容，3 项功能各由 1 个程序实现。

四、实验内容

1、了解教学计算机监控程序功能。

2、指令与系统级模拟系统的使用。

3、BASIC高级语言程序编程体验。

五、实验步骤

1、BASIC程序设计实验

(1) 版本、实现与运行

TEC-2000 BASIC main module v1.00，用TEC-2000教学机的指令系统实现BASIC语言的解释

执行程序，并存放在主存储器ROM区的监控程序之后从0A30单元开始的存储区域。在监控程序启动起来之后，通过G 0A30 命令启动BASIC。




图1. 通过监控程序启动BASIC




图2. 模拟系统路径




图3. 通过模拟系统启动BASIC

(2) 实现的功能

TEC-2000 BASIC解释程序，负责处理与用户的接口，即循环处理执行即时命令和BASIC语言的语句。执行的即时命令包括：

new 清除内存缓冲区已有内容，准备输入一个新的BASIC程序

run 运行已输入的BASIC程序

delete 删除BASIC程序中的指定行号的语句

list 显示输入的一个BASIC程序所有语句

system 退回监控系统

命令和语句输入与执行规则：

即时命令直接输入后就直接开始运行，程序语句前面加行号，如果只输入一个行号后面没有语句，则此行号无效。如果原来存在以此行号为标识的语句，则删除原语句。

输入完一个BASIC程序之后，通过run命令运行这个程序。已经实现的语句包括：

let ， 为变量赋值，该语句名通常可以省略

dim ， 定义数组变量

input ， 从键盘向变量输入新值

print ， 把常量、变量的值显示到计算机的屏幕上

for...next ， 建立一段循环执行的程序段

goto ， 语句跳转到指定的标号之处

gosub ， 调用指定标号的子程序

return ， 子程序返回

end ， 程序结束

表达式处理：

可用的运算符：+(加)，-(减)，*(乘)，/(除)，\ (整除)，^(乘方)，mod(取模)

可用的关系副：>(大于)，>=(大于等于)，<(小于)，<=(小于等于)，<>(不等于)

可用的逻辑运算副：not(取反)，and(与运算)，or(或运算)，xor(异或运算)

可用的函数：sin(正弦)，cos(余弦)，tan(正切)，atan(反正切)，log(10为底的对数)，

exp(10为底的指数) sgn(取数的符号位)，abs(绝对值)，int(取整)，sqr (开平方)

(3) 用到监控程序中的子程序：

057fh: 显示一串字符，056bh: 显示一个字符，0589h: 输入一串字符。

(4) 程序处理的大体思路

对于输入每个语句或命令，是通过检查是否带有行号来区分的，若有，则转一般语句处理部分。否则，转即时命令处理部分。

BSASIC语言的解释执行程序的流程为一循环，前面先把堆栈指针位置记录下来。在循环中，先判断当前语句是否为空，若为空，则结束执行。然后查看当前是否为一新行，或是冒号。若为一个新行，则将下条指令的地址算出，存入相应的内存单元。然后判断是否为一条空语句，如是则转入下个循环。最后检查是哪一条语句，转入相应处理。在每个循

环的最后要检查结束标志是否设置，如设置则结束处理过程。

2、BASIC编程测试

(1) 下面是一个完成整数排序功能的程序，要求首先输入5个参加排序的整数数值，接下来完成对这5个整数的排序操作，并输出最终的排序结果。

```
10  for i=1 to 5
20  input a(i)
30  next i
40  for i=1 to 4
50  for j=i+1 to 5
60  if a(i)>a(j) then b=a(i) : a(i)=a(j) : a(j)=b
70  next j
80  next i
90  for i=1 to 5
100 print a(i)
110 next i
120 end
```

(2) 这是一个求素数的程序，即在指定的数据(100)范围内，找出除了能被1和这个数本身整除之外，不会再被另外的数整除的全部正整数，并将结果显示在计算机屏幕上。

```
10  dim a(100)
20  for i=2 to 100
30  j=i
40  j=j+1
50  if i*j<100 then a(i*j)=1 : goto 40
60  next i
70  for i=2 to 99
80  if a(i)=0 then print i,
90  next i
100 end
```

(3) 这是一个计算正弦三角函数的程序，将0~360度范围内的正弦曲线显示在计算机屏幕上。

```
10  pi=3.14159
20  for i=0 to 20
30  angle=pi*i/10
40  for j=1 to 40+25*sin(angle)
50  print " "
60  next j
70  print "*"
80  next i
90  end
```

(4) 这是一个解决8皇后问题的程序，是在8行*8列的棋盘上，以相互不能“吃子”的方式放进8个皇后棋子，即在任何一个横排上、任何一个竖列上、任何一个对角线的方向上，都不得同时出现两个皇后棋子，把全部可行结果排列出来，并显示在计算机屏幕上。

```
10  dim colstate(7), fdiagstate(14), bdiagstate(14), queenpos(7)
20  i=0 : count=0
30  queenpos(i)=0
40  if colstate(queenpos(i))+fdiagstate(i-queenpos(i)+7)+bdiagstate(i+queenpos(i)) >0 then 170
50  colstate(queenpos(i))=1 : fdiagstate(i-queenpos(i)+7)=1 : bdiagstate(i+queenpos(i))=1
60  if i<7 then i=i+1 : goto 30
70  count=count+1
80  print : print "Result:"; count; ":"
90  j=0
100 k=0
110 if k=queenpos(j) then print "0"; : goto 130
120 print ":";
130 k=k+1 : if k<8 then 110
140 print
150 j=j+1 : if j<8 then 100
160 colstate(queenpos(i))=0 : fdiagstate(i-queenpos(i)+7)=0 : bdiagstate(i+queenpos(i))=0
170 queenpos(i)= queenpos(i)+1 : if queenpos(i)<8 then 40
180 i=i-1 : if i>=0 then 160
190 print : print "Total results:"; count
200 end
```

3. BASIC程序设计实验仿真

运行TEC2kSim模拟器，启动BASIC编辑器，编写BASIC程序。用list命令查看输入的程序，用run命令运行，最后用system命令退出BASIC。






图 4 指令与系统级的模拟系统主界面及主要功能菜单说明

六、实验结果分析与总结

运行上述实验程序，并分析结果。

七、思考题

编写一个在数值100范围内验证歌德巴赫猜想的程序，即任何一个大于2的偶数都等于另外两个素数之和，把验证的结果显示在计算机屏幕上。